

PrácticaJP5

Primitivas e Integrales Definidas.

Aplicaciones Geométricas del Cálculo Integral.

Primitivas e Integrales definidas.

Sage permite el cálculo de integrales de un modo sencillo e intuitivo. Por ejemplo, si lo que queremos es calcular las primitivas de una cierta función f ; es decir, evaluar la integral indefinida $\int f(x) dx$, debemos usar el comando `integrate(f,x)` veamos algunos ejemplos.

```
integral(x*cos(x^2),x)
1/2*sin(x^2)
```

```
integral(1+cos(x),x)
x + sin(x)
```

Otra forma de solicitarle una primitiva a Sage es mediante la orden `f.integrate(x)`.

```
(1/(exp(x)+1)).integrate(x)
x - log(e^x + 1)
```

Observar que en todos los casos Sage está omitiendo la constante de integración.

El teorema fundamental del cálculo integral nos permite asegurar que cualquier función continua en un intervalo, admite primitiva en dicho intervalo. Sin embargo, no es difícil encontrar funciones continuas cuya primitiva no es expresable en términos de funciones elementales (se denomina *función elemental* a cualquier combinación de funciones potenciales, exponenciales, trigonométricas y sus inversas). Uno de los ejemplos más claros es la función gaussiana e^{-x^2} .

```
exp(-x^2).integrate(x)
1/2*sqrt(pi)*erf(x)
```

Sage nos expresa la primitiva de la gaussiana en términos de la denominada función de error

$$\operatorname{erf}(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2} dt,$$

Este ejemplo no es, ni mucho menos, único.

Cuando lo que deseemos sea calcular la integral definida f en $[a,b]$, es decir cuando queramos evaluar $\int_a^b f(x) dx$ la forma de solicitarle a Sage que nos la calcule es `integrate(f,x,a,b)`.

```
integral(x*cos(x^2),x,0,sqrt(pi))
0
```

Como en el caso del cálculo de primitivas la forma tenemos otra forma alternativa de solicitarle a Sage que nos calcule $\int_a^b f(x) dx$ mediante la estructura `f.integrate(x,a,b)`.

```
(1/(exp(x)+1)).integrate(x,0,1)
-log(e + 1) + log(2) + 1
```

```
((1/(exp(x)+1)).integrate(x,0,1)).simplify_log()
log(2/(e + 1)) + 1
```


Veamos algo más elaborado. Estudiamos el tamaño de una integral dependiente de un cierto

parámetro. En concreto nos gustaría saber el comportamiento de las integrales

$$I_k = \int_{-\pi}^{\pi} (\pi - |x|) \cos(kx) dx$$

para valores enteros no negativos de k . Para hacernos una idea de estas funciones comenzamos pintandonos algunos casos particulares.

```
reset(); x,k=var('x,k'); f(k,x)=(pi-abs(x))*cos(x*k)
show(sum(plot(f(k,x),(x,-pi,pi),rgbcolor=hue(1/(8-k)))) for k in [1..7])
```




Como la función a integrar es par, basta calcular su integral en $[0,\pi]$ y multiplicar por dos.

```
g(k)=2*integrate(f(k,x),x,0,pi)
```

```
L=[(k,g(k)) for k in [1..13]]
print(L)
```

```
[(1, 4), (2, 0), (3, 4/9), (4, 0), (5, 4/25), (6, 0), (7, 4/49),
(8, 0), (9, 4/81), (10, 0), (11, 4/121), (12, 0), (13, 4/169)]
```

```
show(list_plot(L,plotjoined=True, color='blue')+ list_plot(L,
color='red'),aspect_ratio=1)
```



Vemos que el valor de la integral va tendiendo hacia cero.

En el caso de que debamos evaluar una integral pero Sage no pueda hacerlo manera exacta, el comando `numerical_integral(f(x),a,b)` nos da una aproximación numérica. Sage nos devolverá como resultado un vector con dos componentes, la primera de ellas es el resultado aproximado y la segunda es una estimación para el error. Si sólo queremos conocer el valor aproximado de la integral basta solicitar a Sage `numerical_integral(f(x),a,b)[0]`, para conocer el error la forma es `numerical_integral(f(x),a,b)[1]`.

```
reset(); x=var('x')
numerical_integral((1+cos(x))/(log(x)+sqrt(1+sin(x))),2,3)
(0.10836134248944476, 1.2030525741107386e-15)
```

```
numerical_integral((1+cos(x))/(log(x)+sqrt(1+sin(x))),2,3)[0]
0.10836134248944476
```


```
numerical_integral((1+cos(x))/(log(x)+sqrt(1+sin(x))),2,3)[1]
1.2030525741107386e-15
```

La curva dada en coordenadas paramétricas por

$$(x(t), y(t)) = \left(\int_0^t \cos(x^2) dx, \int_0^t \sin(x^2) dx \right)$$

se denomina espiral de Cornu y está relacionada con la difracción de la luz. Las integrales que aparecen no pueden ser evaluadas de manera exacta y deben calcularse de manera aproximada. Veamos como Sage nos ayuda a hacernos una idea de la forma de la curva.

```
reset(); x,t,k=var('x,t,k')
PL=[numerical_integral(cos(x^2),0,k/200)[0],
numerical_integral(sin(x^2),0,k/200)[0]] for k in range(0,1000,1)]
show(list_plot(PL,color='red',plotjoined=True),aspect_ratio=1)
```



Integrales impropias.

Sage manipula las integrales impropias considerándolas como cualquier otro tipo de integral definida. Sage devolverá los resultados en determinados casos de integrales impropias convergentes, pero no siempre va a ser posible calcular el valor de la integral, aunque sea cierta la convergencia. Veamos algunos ejemplos.

```
reset(); x=var('x'); integrate(-log(x),x,0,1)
```

1

```
integrate(log(x)/(1+x^2),x,0,infinity)
```


0

Lo que está ocurriendo con la integral anterior es que el valor de la integral en $(0,1)$ es opuesto al de la integral en $(1,\infty)$ (y, por supuesto, ambas son convergentes) y su suma se cancela. Para comprobar esto basta hacer el cambio de variable $x = \frac{1}{t}$,

$$\int_0^1 \frac{\log x}{1+x^2} dx = - \int_\infty^1 \frac{\log \frac{1}{t}}{1+\frac{1}{t^2}} \frac{dt}{t^2} = - \int_1^\infty \frac{\log t}{1+t^2} dt.$$

La gráfica de la función nos permite hacernos una idea de lo que ocurre:

```
plot(log(x)/(1+x^2),x,0,15)
```



P9or supuesto, en el caso de encontrarse con una integral divergente Sage nos advierte de ello.

```
integrate(tan(x),x,0,pi/2)
```

Traceback (click to the left of this block for traceback)

...

ValueError: Integral is divergent.

```
integrate(x^(3/2)/(1+x^2),x,0,infinity)
```

Traceback (click to the left of this block for traceback)

...

ValueError: Integral is divergent.

Para discutir la convergencia de una integral dependiente de un parámetro, Sage nos va cuestionando sobre los valores del parámetro. Para responder a sus preguntas tenemos el comando `assume(...)`. Si queremos introducir modificar el valor del parámetro, antes de un nuevo `assume(...)`, debemos incluir el comando `forget(...)` para que olvide la suposición anterior. Después de cada suposición debemos pedir a Sage que calcule la integral. Por ejemplo, estudiemos la clásica $\int_1^\infty \frac{dx}{x^a}$.

```
a=var('a'); integrate(1/(x^a),x,1,infinity)
```

Traceback (click to the left of this block for traceback)

...

Is a positive, negative, or zero?

```
assume(a<0); integrate(1/(x^a),x,1,infinity)
```

Traceback (click to the left of this block for traceback)

...

ValueError: Integral is divergent.

```
forget(); assume(a>1); integrate(1/(x^a),x,1,infinity)
```

1/(a - 1)


Sage tiene implementada la función Gamma de Euler: $\Gamma(p) = \int_0^\infty e^{-t} t^{p-1} dt$, $\forall p > 0$

Con Sage podemos comprobar la conocida identidad $\Gamma(n+1) = n!$ y que $\Gamma(1/2) = \sqrt{\pi}$:

```
[simplify(gamma(n+1)==factorial(n)) for n in (1..10)]
[True, True, True, True, True, True, True, True, True, True]
gamma(1/2)
sqrt(pi)
```

o hacernos una idea de la representación gráfica de la función Gamma de Euler.

```
plot(gamma(x), x, 0.1, 4)
```



Aplicaciones Geométricas del Cálculo Integral.

Áreas de regiones planas.

El cálculo de áreas de regiones planas puede hacerse con integrales definidas. Sean f y g funciones integrables en el intervalo $[a, b]$, entonces el área de la región limitada por ambas curvas entre las rectas verticales $x = a$ y $x = b$ está dada por la integral definida

$$\int_a^b |f(x) - g(x)| dx.$$

Si una curva es el eje horizontal, por ejemplo, si $g(x) = 0$, el área se reduce a $\int_a^b |f(x)| dx$. Si, además, $f(x)$ es positiva el área coincide con la integral definida en $[a, b]$.

Por otra parte, es claro que si $f(x) \geq g(x)$ en $[a, c]$ y $f(x) \leq g(x)$ en $[c, b]$, entonces el área limitada por ambas curvas en $[a, b]$ es

$$\int_a^c (f(x) - g(x)) dx + \int_c^b (g(x) - f(x)) dx.$$


Ejemplo 1.

Sean $g(x) = x^2 + 2$ y $f(x) = 2x + 5$. Dibujar las gráficas de las dos funciones sobre los mismos ejes y encontrar el área encerrada entre las dos curvas y las rectas $x = 0$ y $x = 6$.

Para dibujar dos funciones f y g y que Sage nos marque la región limitada entre ambas debemos juntar en un `show` el gráfico de una de ellas, por ejemplo f , y en el gráfico de la otra debemos incorporar la orden `fill=f` que le indica a Sage que debe sombrear la región comprendida entre la curva que está pintando y la curva f . Con el comando `fillcolor`

indicamos el color.

```
reset(); x=var('x'); f(x)=2*x+5; g(x)=x^2+2
show(plot(f(x), 0, 6, color='blue')+
plot(g(x), 0, 6, color='green', fill=f(x), fillcolor='red'))
```



```
integrate(abs(f(x)-g(x)), x, 0, 6)
integrate(abs(x^2 - 2*x - 3), x, 0, 6)
```

Sage tiene problemas con integrales con valores absolutos. De otro modo:

```
solve(f(x)-g(x)==0, x)
[x == 3, x == -1]
```


Entre $(0, 3)$ la recta es mayor que la parábola y en $(3, 6)$ al contrario. Así:

```
integrate(f(x)-g(x), x, 0, 3)+integrate(g(x)-f(x), x, 3, 6)
36
```

Ejemplo 2.

Determinar el área de un elipse de semiejes c y d : $\frac{x^2}{c^2} + \frac{y^2}{d^2} = 1$.

```
reset(); x,c,d=var('x,c,d'); h(x,c,d)=d*sqrt(1-x^2/c^2)
show(plot(h(x,5,2), x,-5,5)+plot(-h(x,5,2), x,-5,5, fill=h(x,5,2)),
aspect_ratio=1)
```



A la vista de la simetría basta calcular el área limitada por la función h en el intervalo $[0, c]$ y multiplicar por cuatro el resultado.

```
assume(c>0); 4*integrate(h(x,c,d),x,0,c)
pi*c*d
```

Volúmenes de revolución.

Sea $f(x)$ una función mayor o igual que 0, integrable en $[a, b]$. Supongamos que la región del plano limitada por la gráfica de f y las rectas $y = 0$, $x = a$ y $x = b$ gira alrededor del eje OX . El sólido de revolución que aparece tiene un volumen dado por la fórmula

$$V_{OX} = \pi \int_a^b (f(x))^2 dx.$$

Si en lugar de girar alrededor de OX , giramos la región anterior alrededor del eje OY , entonces el volumen del cuerpo engendrado es

$$V_{OY} = 2\pi \int_a^b x f(x) dx.$$

Ejemplo 1.


Como primer ejemplo obtendremos el volumen de un *toro*. En matemáticas un toro es una figura con forma de rosquilla. Esta figura puede obtener, por ejemplo, como el giro alrededor de OX de la circunferencia $x^2 + (y - A)^2 = B^2$ con $A > B$ (si $A = B$ el agujero de la rosquilla se reduciría a un punto y si $A < B$ no habría agujero). En realidad, obtendremos la superficie girando las funciones $f_1(x) = A + \sqrt{B^2 - x^2}$ y $f_2(x) = A - \sqrt{B^2 - x^2}$ para $x \in [-B, B]$ (El toro se podría obtener también girando alrededor de OY la circunferencia $(x - A)^2 + y^2 = B^2$)

En primer lugar visualizaremos nuestra superficie, para ello debemos observar que si giramos una curva $y = f(x)$ para $x \in [a, b]$, las coordenadas paramétricas de la superficie de revolución obtenida al girar la curva alrededor de OX serán

$$(x, f(x) \cos \theta, f(x) \sin \theta), \quad x \in [a, b], \quad \theta \in [0, 2\pi].$$

Para pintar una superficie en coordenadas paramétricas Sage utiliza la orden `parametric_plot3d((fx(u,v), fy(u,v), fz(u,v)), [u,u1,u2], [v,v1,v2])` a este comando se le pueden añadir todas las opciones de diseño habituales. En nuestro caso vamos a pintar el toro generado por nuestra circunferencia tomando $A = 2$ y $B = 1$.

```
reset(); x,t,A,B=var('x,t,A,B')
show(parametric_plot3d((x,(2+sqrt(1-(x)^2))*cos(t), (2+sqrt(1-(x)^2))*sin(t)), [x,-1,1], [t,0,2*pi])+ parametric_plot3d((x,(2-sqrt(1-(x)^2))*cos(t), (2-sqrt(1-(x)^2))*sin(t)), [x,-1,1], [t,0,2*pi]), aspect_ratio=1)
```



[Get Image](#)

El volumen del toro será el generado por la revolución de la función $f_1(x)$ menos el generado por la función $f_2(x)$. Por tanto

```
assume(B>0); simplify(pi*integrate((A^2+sqrt(B^2-x^2))^2- (A^2-
sqrt(B^2-x^2))^2,x,-B,B))
2*pi^2*A^2*B^2
```


Ejemplo 2.

Calcular el volumen obtenido al girar en torno a OY la región limitada en el intervalo $[0, \pi]$ por las gráficas de $f(x) = \sin x$ y $g(x) = -\sin x$.

Como antes para dibujar la superficie usamos coordenadas paramétricas, la superficie obtenida al girar alrededor de OY una curva $y = f(x)$, para x en el intervalo $[a, b]$, tiene coordenadas

$$(x \cos \theta, f(x), x \sin \theta), \quad a \in [a, b], \quad \theta \in [0, 2\pi].$$

```
reset(); x,t=var('t,x')
show(parametric_plot3d((x*cos(t),sin(x),x*sin(t)), [x,0,pi],
[t,0,2*pi])+ parametric_plot3d((x*cos(t),-sin(x),-x*sin(t)),
[x,0,pi], [t,0,2*pi]), aspect_ratio=1)
```



[Get Image](#)

Por la simetría, calcularemos el volumen generado por f_1 y lo multiplicaremos por dos.

```
2*(2*pi*integrate(x*sin(x),x,0,pi))
4*pi^2
```

Longitud de una curva en coordenadas cartesianas.

Sea $f(x)$ una función definida en $[a, b]$. La longitud del arco de curva definido por la gráfica de esa función es


$$L = \int_a^b \sqrt{1 + (f'(x))^2} dx.$$

Ejemplo 1.

Calcular la longitud de la curva $y = \log(\cos x)$ para $x \in [0, a]$ con $a < \pi/2$

La función sólo está definida en intervalos del tipo $\left[\frac{(2k+1)\pi}{2}, \frac{(2k+3)\pi}{2}\right]$, con k entero. Visualizaremos la curva en el intervalo $[0, \pi/2]$ y sobre ella marcaremos un segmento rojo que será del que supuestamente calculemos la longitud.

```
reset(); x,a=var('x,a'); j(x)=log(cos(x))
show(plot(log(cos(x)),x,0,1.58)+ plot(log(cos(x)),x,0,4*pi/9,
color='red'))
```



```
sqrt(1+(diff(j(x),x))^2).simplify_full()
1/abs(cos(x))
assume(a>0); assume(a<pi/2)
integrate(1/cos(x),x,0,a).simplify_full()
1/2*log(sin(a) + 1) - 1/2*log(abs(sin(a) - 1))
```


Ejemplo 2.

Calcular la longitud del arco de la curva $9y^2 = x(x-3)^2$ limitado por el eje OX .

La curva está formada por las funciones $f_1(x) = \frac{(x-3)\sqrt{x}}{3}$ y $f_2(x) = -\frac{(x-3)\sqrt{x}}{3}$.

Hallamos los cortes con OX y la pintamos:

```
reset(); x=var('x')
f1(x)=(x-3)*sqrt(x)/3; f2(x)=-(x-3)*sqrt(x)/3
solve(f1(x)==0,x); solve(f2(x)==0,x)
[x == 0, x == 3]
[x == 0, x == 3]
plot((f1(x),f2(x)),0,4,aspect_ratio=1)
```



Por la simetría, la longitud pedida será dos veces la de la función f_1 en el intervalo $[0, 3]$.

```
integrate(sqrt(1+(diff(f1(x),x))^2),x,0,3)
2*sqrt(3)
```